Règle de L'Hospital

© Pierre Lantagne (Avril 2018)
Enseignant retraité du Collège de Maisonneuve

Ce document est une révision de celui produit en 2006. L'objectif principal de ce document Maple est de rendre l'élève apte à développer *pas-à-pas* des calculs de limites à l'aide de la règle de L'Hospital pour lever différents types d'indétermination. On abordera, à la fin de ce document, deux exemples de calculs de limites pour lesquelles l'application de la règle de L'Hospital s'avère inefficace. Ce document Maple servira aussi à parfaire la maîtrise du logiciel afin de faire de bons développements mathématiques avec Maple.

Dans ce document, nous n'emploirons les règles de calculs de limites de la sous-bibliothèque *Student*[*Calculus1*] seulement pour compléter le développement pas-à-pas. Seule la règle *lhopital* de la macro-commande *Rule* ne sera pas utilisée afin de « contrôler » l'application de la règle de l'Hospital. Les développements *pas-à-pas* proposés dans ce document des calculs de limites sont faits avec une approche pédagogique se rapprochant aux calculs exigés en classe dans un développement manus scriptus.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
```

```
> with(plots,display,pointplot,setoptions);
  setoptions(axesfont=[TIMES,ROMAN,8],size=[300,300]):
```

$$[display, pointplot, setoptions] \quad (1)$$

```
> with(Student[Calculus1],Rule);
  infolevel[Student[Calculus1]]:=1:
```

$$[Rule] \quad (2)$$

Indétermination de la forme 0/0

Exemple 1

Soit la fonction f définie par $f(x) = \dfrac{8^x - 5^x}{5x}$, dont le domaine est $\mathbb{R} \setminus \{0\}$.

```
> f:=x->(8^x-5^x)/(5*x);
```

$$f := x \mapsto \frac{8^x - 5^x}{5 \cdot x} \quad (3)$$

Comme le dénominateur de $f(x)$ s'annule en $x = 0$, la fonction f n'est donc pas continue en $x = 0$. Pour toutes valeurs réelles x au voisinage de 0, le numérateur et le dénominateur de $f(x)$ sont toujours définis et non nuls. Visualisons cette discontinuité en $x = 0$ afin d'en connaître la nature. Explorons le tracé de la fonction f dans un voisinage de $x = 0$.

```
> Tracé:=plot([x,f(x),x=-1..1],color=orange):
  Tracé;
```



Zoomons ce tracé vers l'avant pour mieux l'explorer au voisinage de $x = 0$.

```
> display(Tracé,view=[-0.1..0.1,0.08..0.12]);
```



Le graphique de la fonction *f* semble toujours montrer que son tracé coupe l'axe des *y*. Nous savons pertinemment qu'il n'en est rien car la fonction *f* n'est pas définie en $x = 0$. Ce zoom vers l'avant suggère assez clairement qu'en $x = 0$, il se présente une ***discontinuité réductible***.

Pour le montrer, évaluons directement la limite $\lim_{x \to 0} \frac{8^x - 5^x}{5x}$ avec la macro-commande *limit*.

```
> Limit(f(x),x=0)=limit(f(x),x=0);
```

$$\lim_{x \to 0} \frac{8^x - 5^x}{5x} = \frac{3\ln(2)}{5} - \frac{\ln(5)}{5} \quad \textbf{(4)}$$

Comme on pouvait s'y attendre, Maple a calculée cette limite sans difficulté, « à sa façon ». Comme la limite existe en $x = 0$ (la fonction *f* est toujours définie dans n'importe quel voisinage de $x = 0$), cela établi qu'il s'agit bien d'une **discontinuité réductible en** $x = 0$. Dans le cas d'une telle discontinuité, seule une approche analytique permet de le prouver. Un point n'ayant aucune dimension, on ne peut pas illustrer ce type de discontinuité car les points générés par la macro-commande *plot* sont ensuite reliés entre-eux tout en ignorant les valeurs d'abcisses qui ne possèdent pas d'image.

Calculons autrement cette limite. Appliquons pas-à-pas les règles de calcul des limites.

```
> Rule[`/`](Limit(f(x),x=0));
Creating problem #1
```

```
Rule [quotient] does not apply
```

$$\lim_{x\to 0}\frac{8^x-5^x}{5x}=\lim_{x\to 0}\frac{8^x-5^x}{5x} \quad \textbf{(5)}$$

Analysons pouquoi la règle *quotient* ne s'applique pas. Évaluons séparément la limite du numérateur et la limite du dénominateur directement.

```
> Limit(numer(f(x)),x=0)=limit(numer(f(x)),x=0);
  Limit(denom(f(x)),x=0)=limit(denom(f(x)),x=0);
```

$$\lim_{x\to 0}\left(8^x-5^x\right)=0$$

$$\lim_{x\to 0}5x=0 \quad \textbf{(6)}$$

L'application de la règle *quotient* n'a pas indentifiée cette indétermination du type $\frac{0}{0}$ mais seulement que la limite du dénominateur est égale à 0, d'où le message que la règle *quotient* ne s'appliquait pas.

Afin d'appliquer la règle de L'Hospital, formons le quotient des dérivées.

```
> d:=x->Diff(numer(f(x)),x)/Diff(denom(f(x)),x);
```

$$d := x\to\frac{\frac{d}{dx}\,numer(f(x))}{\frac{d}{dx}\,denom(f(x))} \quad \textbf{(7)}$$

Posons ensuite la limite en $x=0$ du quotient des dérivées.

```
> Limit(f(x),x=0)=Limit(d(x),x=0);
```

$$\lim_{x\to 0}\frac{8^x-5^x}{5x}=\lim_{x\to 0}\frac{\frac{d}{dx}\left(8^x-5^x\right)}{\frac{d}{dx}\left(5x\right)} \quad \textbf{(8)}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d(x))),x=0);
```

$$=\lim_{x\to 0}\left(\frac{3\,8^x\ln(2)}{5}-\frac{5^x\ln(5)}{5}\right) \quad \textbf{(9)}$$

Appliquons les règles du calcul des limites.

```
> Rule[`-`]((9));
Creating problem #2 from the right hand side of this equation
```

$$\lim_{x\to 0}\left(\frac{3\,8^x\ln(2)}{5}-\frac{5^x\ln(5)}{5}\right)=\left(\lim_{x\to 0}\frac{3\,8^x\ln(2)}{5}\right)+\left(\lim_{x\to 0}-\frac{5^x\ln(5)}{5}\right) \quad \textbf{(10)}$$

```
> Rule[`c*`]((10));
```

$$\lim_{x\to 0}\left(\frac{3\,8^x\ln(2)}{5}-\frac{5^x\ln(5)}{5}\right)=\frac{3\ln(2)\left(\lim_{x\to 0}8^x\right)}{5}+\left(\lim_{x\to 0}-\frac{5^x\ln(5)}{5}\right) \quad \textbf{(11)}$$

```
> Rule[`c*`]((11));
```

$$\lim_{x\to 0}\left(\frac{3\,8^x\ln(2)}{5}-\frac{5^x\ln(5)}{5}\right)=\frac{3\ln(2)\left(\lim_{x\to 0}8^x\right)}{5}-\frac{\ln(5)\left(\lim_{x\to 0}5^x\right)}{5} \qquad (12)$$

Reste à évaluer directement ces dernières limites.

```
> Rule[`^`]((12));
```

$$\lim_{x\to 0}\left(\frac{3\,8^x\ln(2)}{5}-\frac{5^x\ln(5)}{5}\right)=\frac{3\ln(2)\left(\lim_{x\to 0}8^x\right)}{5}-\frac{\ln(5)\left(\lim_{x\to 0}5\right)^{\lim_{x\to 0}x}}{5} \qquad (13)$$

```
> Rule[`^`]((13));
```

$$\lim_{x\to 0}\left(\frac{3\,8^x\ln(2)}{5}-\frac{5^x\ln(5)}{5}\right)=\frac{3\ln(2)\left(\lim_{x\to 0}8^x\right)}{5}-\frac{\ln(5)}{5} \qquad (14)$$

```
> Rule[`^`]((14));
```

$$\lim_{x\to 0}\left(\frac{3\,8^x\ln(2)}{5}-\frac{5^x\ln(5)}{5}\right)=\frac{3\ln(2)\left(\lim_{x\to 0}8\right)^{\lim_{x\to 0}x}}{5}-\frac{\ln(5)}{5} \qquad (15)$$

```
> Rule[`^`]((15));
```

$$\lim_{x\to 0}\left(\frac{3\,8^x\ln(2)}{5}-\frac{5^x\ln(5)}{5}\right)=\frac{3\ln(2)}{5}-\frac{\ln(5)}{5} \qquad (16)$$

L'indétermination est donc levée. Mais, avant de formuler la réponse dans une zone de texte, simplifions le résultat précédent à l'aide des propriétés des logarithmes.

Réponse: $\lim_{x\to 0}\frac{(8^x-5^x)}{5x}=\frac{\ln\left(\frac{8}{5}\right)}{5}$.

Voyons la simplification que donne Maple.

```
> (combine((16)));
```

$$\lim_{x\to 0}\left(\frac{3\,8^x\ln(2)}{5}-\frac{5^x\ln(5)}{5}\right)=\frac{3\ln\left(\frac{2\,5^{2/3}}{5}\right)}{5} \qquad (17)$$

Il est instructif d'explorer le tracé de la fonction f et celui du quotient des dérivées au voisinage de $x=0$ afin de constater que les deux tracés possèdent effectivement la même limite en $x=0$.

```
> plot([[x,f(x),x=-1..1],[x,value(d(x)),x=-1..1]],color=[orange,
  navy],legend=["y=f(x)","y=d(x)"],view=[-0.8..0.8,-0.1..0.5]);
```



Exemple 2

Soit la fonction *f* définie par $f(t) = \dfrac{e^t - e^{-t} - 2t}{t - \sin(t)}$ dont le domaine est $\mathbb{R} \setminus \{0\}$.

```
> f:=t->(exp(t)-exp(-t)-2*t)/(t-sin(t));
```

$$f := t \mapsto \frac{e^t - e^{-t} - 2 \cdot t}{t - \sin(t)} \tag{18}$$

```
> solve(denom(f(t))=0,{t});
```

$$\{t = 0\} \tag{19}$$

Le dénominateur de *f*(*t*) s'annulant en $t = 0$, la fonction *f* n'est donc pas continue en $t = 0$. Pour toutes valeurs réelles *t* au voisinage de 0, le numérateur et le dénominateur de *f*(*t*) sont toujours définis et non nuls. Explorons le tracé de la fonction *f* dans un voisinage de $t = 0$ afin de connaître la nature de cette discontinuité en $t = 0$.

Obtenons le tracé de la fonction *f* dans un voisinage de $t = 0$.

```
> Tracé:=plot([t,f(t),t=-1..1],color=orange):
  Tracé;
```



Encore une fois, le graphique suggère un tracé continu en $t = 0$.

Zoomons vers l'avant ce tracé dans le voisinage de $t = 0$.

```
> display(Tracé,view=[-0.2..0.2,1.9..2.1]);
```



Par un calcul, montrons que la limite existe en $t = 0$. On pourra, de ce fait, conclure qu'il s'agit d'une discontinuité réductible. Évaluons la limite directement avec la macro-commande *limit*.

```
> Limit(f(t),t=0)=limit(f(t),t=0);
```

$$\lim_{t \to 0} \frac{e^{t} - e^{-t} - 2t}{t - \sin(t)} = 2 \qquad \textbf{(20)}$$

En $t = 0$, il se présente effectivement une **discontinuité réductible**.

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites.

```
> Rule[`/`](Limit(f(t),t=0));
Creating problem #3

This problem is an indeterminate form
```

$$\lim_{t \to 0} \frac{e^{t} - e^{-t} - 2t}{t - \sin(t)} = \lim_{t \to 0} \frac{e^{t} - e^{-t} - 2t}{t - \sin(t)} \qquad \textbf{(21)}$$

Analysons cette forme d'indétermination. Évaluons séparément la limite du numérateur et la limite du dénominateur directement avec la macro-commande *limit*.

```
> Limit(numer(f(t)),t=0)=limit(numer(f(t)),t=0);
  Limit(denom(f(t)),t=0)=limit(denom(f(t)),t=0);
```

$$\lim_{t \to 0} \left(-e^{t} + e^{-t} + 2t \right) = 0$$

$$\lim_{t \to 0} \left(-t + \sin(t) \right) = 0 \qquad \textbf{(22)}$$

Nous avons donc une indétermination du type $\frac{0}{0}$.

Afin d'appliquer la règle de L'Hospital, formons le quotient des dérivées.

```
> d:=t->Diff(numer(f(t)),t)/Diff(denom(f(t)),t);
```

$$d := t \rightarrow \frac{\frac{d}{dt} \mathit{numer}(f(t))}{\frac{d}{dt} \mathit{denom}(f(t))} \qquad \textbf{(23)}$$

Appliquons le règle de L'Hospital. Développons la limite en $x = 0$ du quotient des dérivées.

```
> Limit(f(t),t=0)=Limit(d(t),t=0);
```

$$\lim_{t \to 0} \frac{e^t - e^{-t} - 2t}{t - \sin(t)} = \lim_{t \to 0} \frac{\frac{d}{dt}\left(-e^t + e^{-t} + 2t\right)}{\frac{d}{dt}\left(-t + \sin(t)\right)} \quad \textbf{(24)}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d(t))),t=0);
```

$$= \lim_{t \to 0}\left(-\frac{e^t + e^{-t} - 2}{-1 + \cos(t)}\right) \quad \textbf{(25)}$$

Appliquons les règles de calcul des limites.

```
> Rule[`/`]((25));
Creating problem #4 from the right hand side of this equation
Rule [quotient] does not apply
```

$$\lim_{t \to 0}\left(-\frac{e^t + e^{-t} - 2}{-1 + \cos(t)}\right) = \lim_{t \to 0}\left(-\frac{e^t + e^{-t} - 2}{-1 + \cos(t)}\right) \quad \textbf{(26)}$$

La règle *quotient* ne s'applique pas. En effet, la limite au dénominateur est égale à 0: $-1 + \cos(t) \to 0$ lorsque $t \to 0$.

Évaluons séparément les limites du numérateur et du dénominateur.

```
> Limit(numer(value(d(t))),t=0)=limit(numer(value(d(t))),t=0);
  Limit(denom(value(d(t))),t=0)=limit(denom(value(d(t))),t=0);
```

$$\lim_{t \to 0}\left(-e^t - e^{-t} + 2\right) = 0$$

$$\lim_{t \to 0}\left(-1 + \cos(t)\right) = 0 \quad \textbf{(27)}$$

Nous avons donc encore une indétermination du type $\frac{0}{0}$.

Appliquons de nouveau la règle de L'Hospital. Formons de nouveau le quotient des dérivées.

```
> d2:=t->Diff(numer(value(d(t))),t)/Diff(denom(value(d(t))),t);
```

$$d2 := t \to \frac{\frac{d}{dt}\, numer(value(d(t)))}{\frac{d}{dt}\, denom(value(d(t)))} \quad \textbf{(28)}$$

Appliquons le règle de L'Hospital.

```
> Limit(f(t),t=0)=Limit(d2(t),t=0);
```

$$\lim_{t \to 0} \frac{e^t - e^{-t} - 2t}{t - \sin(t)} = \lim_{t \to 0} \frac{\frac{d}{dt}\left(-e^t - e^{-t} + 2\right)}{\frac{d}{dt}\left(-1 + \cos(t)\right)} \quad \textbf{(29)}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d2(t))),t=0);
```

(30)

$$= \lim_{t \to 0} \frac{e^t - e^{-t}}{\sin(t)} \quad \mathbf{(30)}$$

Appliquons les règles de calcul des limites.

```
> Rule[`/`]((30));
Creating problem #5 from the right hand side of this equation
This problem is an indeterminate form
```

$$\lim_{t \to 0} \frac{e^t - e^{-t}}{\sin(t)} = \lim_{t \to 0} \frac{e^t - e^{-t}}{\sin(t)} \quad \mathbf{(31)}$$

Encore une fois, Maple a identifié une forme d'indétermination. C'est clair qu'il s'agit de la forme $\frac{0}{0}$.

Évaluons séparément les limites du numérateur et du dénominateur.

```
> Limit(numer(value(d2(t))),t=0)=limit(numer(value(d2(t))),t=0);
  Limit(denom(value(d2(t))),t=0)=limit(denom(value(d2(t))),t=0);
```

$$\lim_{t \to 0} \left(e^t - e^{-t}\right) = 0$$

$$\lim_{t \to 0} \sin(t) = 0 \quad \mathbf{(32)}$$

Appliquons une troisième fois la règle de L'Hospital. On constatera que la dérivée du sinus au dénominateur, va, cette fois, nous permettre de lever l'indétermination.

Formons le quotient des dérivées.

```
> d3:=t->Diff(numer(value(d2(t))),t)/Diff(denom(value(d2(t))),t);
```

$$d3 := t \to \frac{\frac{d}{dt}\, numer(value(d2(t)))}{\frac{d}{dt}\, denom(value(d2(t)))} \quad \mathbf{(33)}$$

Appliquons la règle de L'Hospital.

```
> Limit(f(t),t=0)=Limit(d3(t),t=0);
```

$$\lim_{t \to 0} \frac{e^t - e^{-t} - 2t}{t - \sin(t)} = \lim_{t \to 0} \frac{\frac{d}{dt}\left(e^t - e^{-t}\right)}{\frac{d}{dt}\sin(t)} \quad \mathbf{(34)}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d3(t))),t=0);
```

$$= \lim_{t \to 0} \frac{e^t + e^{-t}}{\cos(t)} \quad \mathbf{(35)}$$

Appliquons les règles de calcul des limites.

```
> Rule[`/`]((35));
Creating problem #6 from the right hand side of this equation
```

$$\lim_{t \to 0} \frac{e^t + e^{-t}}{\cos(t)} = \frac{\lim_{t \to 0}\left(e^t + e^{-t}\right)}{\lim_{t \to 0}\cos(t)} \quad \mathbf{(36)}$$

Évaluons ces limites.

```
> Rule[`+`]((36));
```

$$\lim_{t\to 0}\frac{e^{t}+e^{-t}}{\cos(t)}=\frac{\left(\lim_{t\to 0}e^{t}\right)+\left(\lim_{t\to 0}e^{-t}\right)}{\lim_{t\to 0}\cos(t)} \tag{37}$$

```
> Rule[cos]((37));
```

$$\lim_{t\to 0}\frac{e^{t}+e^{-t}}{\cos(t)}=\left(\lim_{t\to 0}e^{t}\right)+\left(\lim_{t\to 0}e^{-t}\right) \tag{38}$$

```
> Rule[exp]((38));
```

$$\lim_{t\to 0}\frac{e^{t}+e^{-t}}{\cos(t)}=1+\left(\lim_{t\to 0}e^{-t}\right) \tag{39}$$

```
> Rule[exp]((39));
```

$$\lim_{t\to 0}\frac{e^{t}+e^{-t}}{\cos(t)}=1+e^{\lim_{t\to 0}(-t)} \tag{40}$$

```
> Rule[`c*`]((40));
```

$$\lim_{t\to 0}\frac{e^{t}+e^{-t}}{\cos(t)}=1+e^{-\left(\lim_{t\to 0}t\right)} \tag{41}$$

```
> Rule[identity]((41));
```

$$\lim_{t\to 0}\frac{e^{t}+e^{-t}}{\cos(t)}=2 \tag{42}$$

L'indétermination est donc finalement levée après trois applications successives de la règle de L'Hospital.

Réponse: $\lim_{t\to 0}\dfrac{e^{t}-e^{-t}-2t}{t-\sin(t)}=2$.

Explorons le tracé de la fonction f et ceux des quotients des dérivées au voisinage de $t=0$ afin de constater que tous les tracés possèdent effectivement la même limite en $t=0$.

```
> plot([[t,f(t),t=-0.2..0.2],[t,value(d(t)),t=-0.2..0.2],[t,value(d2
  (t)),t=-0.2..0.2],[t,value(d3(t)),t=-0.2..0.2]],color=[orange,
  navy,khaki,magenta],legend=["y=f(t)","y=d(t)","y=d2(t)","y=d3(t)
  "],view=[-0.2..0.2,1.98..2.04]);
```



Exemple 3

Soit la fonction f définie par $f(x) = \dfrac{\sin(3x)}{1-\cos(4x)}$ dont le domaine est $\mathbb{R} \setminus \left\{ x \,\middle|\, x = \dfrac{k\pi}{2}, k \in \mathbb{Z} \right\}$

```
> f:=x->sin(3*x)/(1-cos(4*x));
```

$$f := x \mapsto \frac{\sin(3 \cdot x)}{1-\cos(4 \cdot x)} \tag{43}$$

```
> _EnvAllSolutions:=true:
  solve(denom(f(x))=0,{x});
```

$$\left\{ x = \frac{\pi\_Z1}{2} \right\} \tag{44}$$

Le dénominateur s'annule en $x = \dfrac{k\pi}{2}$, $k \in \mathbb{Z}$. En particulier, lorsque $k = 0$, on déduit que la fonction f n'est pas continue en $x = 0$. Alors, pour toutes valeurs réelles x au voisinage de 0, le numérateur et le dénominateur de $f(x)$ sont toujours définis et non nuls. Visualisons cette discontinuité en $x = 0$ afin d'en connaître la nature. Explorons le tracé de la fonction f dans un voisinage de $x = 0$.

Obtenons le tracé de la fonction f dans un voisinage de $x = 0$.

```
> Tracé:=plot([x,f(x),x=-1..1],color=orange,
        discont=true,view=[-0.5..0.5,-10..10]):
  Tracé;
```



En $x = 0$, il se présente une **discontinuité par passage à l'infini**. Montrons-le en calculant directement les limites à gauche et à droite.

```
> Limit(f(x),x=0,left)=limit(f(x),x=0,left);
```

```
Limit(f(x),x=0,right)=limit(f(x),x=0,right);
```

$$\lim_{x \to 0-} \frac{\sin(3x)}{1-\cos(4x)} = -\infty$$

$$\lim_{x \to 0+} \frac{\sin(3x)}{1-\cos(4x)} = \infty \quad \textbf{(45)}$$

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites. Nous allons nous retreindre au calcul de la limite à droite seulement.

Appliquons les règles de calcul des limites.

```
> Rule[`/`](Limit(f(x),x=0,right));
Creating problem #7

This problem is an indeterminate form
```

$$\lim_{x \to 0+} \frac{\sin(3x)}{1-\cos(4x)} = \lim_{x \to 0+} \frac{\sin(3x)}{1-\cos(4x)} \quad \textbf{(46)}$$

Analysons la forme d'indétermination. Évaluons séparément la limite du numérateur et la limite du dénominateur directement avec la macro-commande *limit*.

```
> Limit(numer(f(x)),x=0)=limit(numer(f(x)),x=0);
  Limit(denom(f(x)),x=0)=limit(denom(f(x)),x=0);
```

$$\lim_{x \to 0} (-\sin(3x)) = 0$$

$$\lim_{x \to 0} (-1+\cos(4x)) = 0 \quad \textbf{(47)}$$

Nous avons donc une indétermination du type $\frac{0}{0}$.

Afin d'appliquer la règle de L'Hospital, formons le quotient des dérivées.

```
> d:=x->Diff(numer(f(x)),x)/Diff(denom(f(x)),x);
```

$$d := x \to \frac{\frac{d}{dx}\, numer(f(x))}{\frac{d}{dx}\, denom(f(x))} \quad \textbf{(48)}$$

Appliquons le règle de L'Hospital. Développons la limite à droite en $x = 0$ du quotient des dérivées.

```
> Limit(f(x),x=0,right)=Limit(d(x),x=0,right);
```

$$\lim_{x \to 0+} \frac{\sin(3x)}{1-\cos(4x)} = \lim_{x \to 0+} \frac{\frac{d}{dx}(-\sin(3x))}{\frac{d}{dx}(-1+\cos(4x))} \quad \textbf{(49)}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal*.

```
> ``=Limit(normal(value(d(x))),x=0,right);
```

$$= \lim_{x \to 0+} \frac{3\cos(3x)}{4\sin(4x)} \quad \textbf{(50)}$$

Appliquons les règles de calcul des limites.

```
> Rule[`/`]((50));
Creating problem #8 from the right hand side of this equation
Rule [quotient] does not apply
```

$$\lim_{x \to 0+} \frac{3\cos(3x)}{4\sin(4x)} = \lim_{x \to 0+} \frac{3\cos(3x)}{4\sin(4x)} \tag{51}$$

La règle quotient ne s'applique pas. En effet, la limite au dénominateur est égale à 0: $4\sin(4x) \to 0$ lorsque $x \to 0^+$ tandis que la limite numérateur est égale à 3: $3\cos(3x) \to 3$ lorsque $x \to 0^+$.
Évaluons séparément la limite du numérateur et la limite du dénominateur directement avec la macro-commande *limit*.

```
> Limit(numer(normal(value(d(x)))),x=0,right)=limit(numer(normal
  (value(d(x)))),x=0,right);
  Limit(denom(normal(value(d(x)))),x=0,right)=limit(denom(normal
  (value(d(x)))),x=0,right);
```

$$\lim_{x \to 0+} 3\cos(3x) = 3$$

$$\lim_{x \to 0+} 4\sin(4x) = 0 \tag{52}$$

Ce n'est pas un cas d'indétermination du type $\frac{0}{0}$. C'est plutôt un cas d'exception du type $\frac{K}{0}$, $K \neq 0$ $\left(\frac{3^-}{0^+} \to \infty\right)$. Appliquons la règle *dividebyzero*.

```
> Rule[dividebyzero]((51));
```

$$\lim_{x \to 0+} \frac{3\cos(3x)}{4\sin(4x)} = \infty \tag{53}$$

L'indétermination est donc levée, la limite $\lim_{x \to 0+} \frac{\sin(3x)}{1-\cos(4x)}$ n'existe pas.

Donc, $\lim_{x \to 0} \frac{\sin(3x)}{1-\cos(4x)}$ n'existe pas

Indétermination de la forme $\pm\infty/\pm\infty$

Exemple 4

Soit la fonction *f* définie par $f(x) = \frac{\ln(x^2)}{\ln(1+x)}$ dont le domaine est $]-1,\infty] \setminus \{0\}$. Découvrons le comportement de la fonction *f* lorsque x augmente indéfiniment, c'est-à-dire lorsque $x \to \infty$.

```
> f:=x->ln(x^2)/ln(1+x);
```

$$f := x \mapsto \frac{\ln(x^2)}{\ln(x+1)} \tag{54}$$

```
> plot([x,f(x),x=-1..20],color=orange,view=[-3..20,-5..5]);
```



Le tracé de la fonction *f* suggère un comportement horizontalement asymptotique, $f(x) \to 2^-$ lorsque $x \to \infty$. Prouvons-le en calculant directement la limite de la fonction *f* lorsque $x \to \infty$.

```
> Limit(f(x),x=infinity)=limit(f(x),x=infinity);
```

$$\lim_{x \to \infty} \frac{\ln(x^2)}{\ln(x+1)} = 2 \tag{55}$$

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites.

Appliquons les règles de calcul des limites.

```
> Rule[`/`](Limit(f(x),x=infinity));
Creating problem #9

This problem is an indeterminate form
```

$$\lim_{x \to \infty} \frac{\ln(x^2)}{\ln(x+1)} = \lim_{x \to \infty} \frac{\ln(x^2)}{\ln(x+1)} \tag{56}$$

Maple a donné pour résultat une indétermination. Analysons cette réponse. Évaluons séparement la limite du numérateur et la limite du dénominateur directement.

```
> Limit(numer(f(x)),x=infinity)=limit(numer(f(x)),x=infinity);
  Limit(denom(f(x)),x=infinity)=limit(denom(f(x)),x=infinity);
```

$$\lim_{x \to \infty} \ln(x^2) = \infty$$

$$\lim_{x \to \infty} \ln(x+1) = \infty \tag{57}$$

On a donc une indétermination du type $\frac{\infty}{\infty}$.

Afin d'appliquer la règle de L'Hospital, formons le quotient des dérivées.

```
> d:=x->Diff(numer(f(x)),x)/Diff(denom(f(x)),x);
```

$$d := x \to \frac{\frac{d}{dx}\, numer(f(x))}{\frac{d}{dx}\, denom(f(x))} \tag{58}$$

Appliquons la règle de L'Hospital. Posons la limite du quotient des dérivées lorsque $x \to \infty$.

```
> Limit(f(x),x=infinity)=Limit(d(x),x=infinity);
```

(59)

$$\lim_{x \to \infty} \frac{\ln(x^2)}{\ln(x+1)} = \lim_{x \to \infty} \frac{\frac{d}{dx} \ln(x^2)}{\frac{d}{dx} \ln(x+1)} \quad \textbf{(59)}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal*.

```
> ``=Limit(normal(value(d(x))),x=infinity);
```

$$= \lim_{x \to \infty} \frac{2(x+1)}{x} \quad \textbf{(60)}$$

Effectuons la division avant d'évaluer la limite.

```
> ``=Limit(expand(value(d(x))),x=infinity);
```

$$= \lim_{x \to \infty} \left(2 + \frac{2}{x}\right) \quad \textbf{(61)}$$

Appliquons les règles de calcul des limites.

```
> Rule[`+`]((61));
Creating problem #10 from the right hand side of this equation
```

$$\lim_{x \to \infty} \left(2 + \frac{2}{x}\right) = \left(\lim_{x \to \infty} 2\right) + \left(\lim_{x \to \infty} \frac{2}{x}\right) \quad \textbf{(62)}$$

```
> Rule[constant]((62));
```

$$\lim_{x \to \infty} \left(2 + \frac{2}{x}\right) = 2 + \left(\lim_{x \to \infty} \frac{2}{x}\right) \quad \textbf{(63)}$$

```
> Rule[`c*`]((63));
```

$$\lim_{x \to \infty} \left(2 + \frac{2}{x}\right) = 2 + 2\left(\lim_{x \to \infty} \frac{1}{x}\right) \quad \textbf{(64)}$$

```
> Rule[`^`]((64));
```

$$\lim_{x \to \infty} \left(2 + \frac{2}{x}\right) = 2 + \frac{2}{\lim_{x \to \infty} x} \quad \textbf{(65)}$$

```
> Rule[identity]((65));
```

$$\lim_{x \to \infty} \left(2 + \frac{2}{x}\right) = 2 \quad \textbf{(66)}$$

L'indétermination a donc été levée.

Réponse: $\lim_{x \to \infty} \frac{\ln(x^2)}{\ln(x+1)} = 2.$

Explorons le tracé de la fonction *f* et celui du quotient des dérivées au voisinage de l'infini afin de constater que les deux tracés possèdent effectivement le même comportement horizontalement asymptotique à l'infini.

```
> plot([[x,f(x),x=-1..10],[x,value(d(x)),x=-1..10]],color=[orange,
  navy],legend=["y=f(x)","y=d(x)"],view=[-3..10,-5..5]);
```



Exemple 5

Soit la fonction *f* définie par $f(x) = \dfrac{\tan(5x)}{\tan(3x)}$. Explorons le tracé de fonction *f* dans un voisinage de $x = \dfrac{\pi}{2}$.

```
> f:=x->tan(5*x)/tan(3*x);
```

$$f := x \mapsto \frac{\tan(5 \cdot x)}{\tan(3 \cdot x)} \tag{67}$$

```
> plot([x,f(x),x=Pi/3..2*Pi/3.1],color=orange,tickmarks = [spacing(
  (1/8)*Pi), default],view=[Pi/3..2*Pi/3,-1..1]);
```



La fonction *f* n'étant pas définie lorsque $x = \dfrac{\pi}{2}$, la fonction *f* semble donc présenter une discontinuité réductible en $x = \dfrac{\pi}{2}$ (1,5708 approx.). Montrons-le en évaluant directement la limite appropriée avec la macro-commande *limit*.

```
> Limit(f(x),x=Pi/2)=limit(f(x),x=Pi/2);
```

$$\lim_{x \to \frac{\pi}{2}} \frac{\tan(5x)}{\tan(3x)} = \frac{3}{5} \tag{68}$$

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites. Développons la limite d'un quotient.

```
> Rule[`/`](Limit(f(x),x=Pi/2));
Creating problem #11

This problem is an indeterminate form
```

$$\lim_{x \to \frac{\pi}{2}} \frac{\tan(5x)}{\tan(3x)} = \lim_{x \to \frac{\pi}{2}} \frac{\tan(5x)}{\tan(3x)} \quad (69)$$

Maple a identifié une indétermination. Analysons cette réponse en évaluant séparément la limite du numérateur et la limite du dénominateur directement. Nous devons considérer séparément les limites à gauche et à droite compte tenu des voisinages $x = \frac{5\pi}{2}$ et $x = \frac{3\pi}{2}$ de chaque fonction tangente au numérateur et au dénominateur. En effet, rappelons-nous le tracé de la fonction tangente au voisinage de $x = \frac{3\pi}{2}$ et au voisinage de $x = \frac{5\pi}{2}$.

```
> Tracé:=plot([x,tan(x),x=0..3*Pi],discont=true,color=orange,view=
  [0..3*Pi,-5..5]):
  Asymptotes:=seq(plot([(2*k+1)*Pi/2,y,y=-6..6],color=khaki,
  linestyle=4),k=0..3):
  display(Tracé,Asymptotes,tickmarks = [spacing((1/2)*Pi), default])
  ;
```



Considérons d'abord la limite à gauche en $x = \frac{\pi}{2}$.

```
> limit(numer(f(x)),x=Pi/2,left);
  limit(denom(f(x)),x=Pi/2,left);
```

$$\infty$$

$$\infty \quad (70)$$

Puis la limite à droite en $x = \frac{\pi}{2}$.

```
> limit(numer(f(x)),x=Pi/2,right);
  limit(denom(f(x)),x=Pi/2,right);
```

$$-\infty$$

$$-\infty \quad (71)$$

Dans le premier cas, on a une indétermination de la forme $\frac{\infty}{\infty}$ et dans le second cas, une indétermination de la forme $\frac{-\infty}{-\infty}$.

Pour appliquer efficacement la règle de L'Hospital, transformons les tangentes en sinus et cosinus.

```
> f_bis:=x->convert(f(x),sincos);
```

$$f\_bis := x \mapsto convert(f(x), sincos) \tag{72}$$

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites.

```
> Limit(f(x),x=Pi/2)=Limit(f_bis(x),x=Pi/2);
```

$$\lim_{x \to \frac{\pi}{2}} \frac{\tan(5x)}{\tan(3x)} = \lim_{x \to \frac{\pi}{2}} \frac{\sin(5x)\cos(3x)}{\cos(5x)\sin(3x)} \tag{73}$$

Appliquons les règles de calcul des limites.

```
> Rule[`/`]((73));
Creating problem #12 from the right hand side of this equation
```

$$\lim_{x \to \frac{\pi}{2}} \frac{\sin(5x)\cos(3x)}{\cos(5x)\sin(3x)} = \frac{\left(\lim_{x \to \frac{\pi}{2}} \frac{\cos(3x)}{\cos(5x)}\right)\left(\lim_{x \to \frac{\pi}{2}} \sin(5x)\right)}{\lim_{x \to \frac{\pi}{2}} \sin(3x)} \tag{74}$$

La règle *quotient* a scindé la limite en mettant en évidence la limite $\lim_{x \to \frac{\pi}{2}} \frac{\cos(3x)}{\cos(5x)}$. Cette limite est une indétermination $\frac{0}{0}$ Rappelons que la macro-commande *Rule* ne scinde pas les formes indéterminées lorsque les règles *sum*, *difference*, *product*, *quotient* and *devidebyzero* sont appliquées.

```
> Rule[sin]((74));
```

$$\lim_{x \to \frac{\pi}{2}} \frac{\sin(5x)\cos(3x)}{\cos(5x)\sin(3x)} = \frac{\left(\lim_{x \to \frac{\pi}{2}} \frac{\cos(3x)}{\cos(5x)}\right)\sin\left(\lim_{x \to \frac{\pi}{2}} 5x\right)}{\lim_{x \to \frac{\pi}{2}} \sin(3x)} \tag{75}$$

```
> Rule[`c*`]((75));
```

$$\lim_{x \to \frac{\pi}{2}} \frac{\sin(5x)\cos(3x)}{\cos(5x)\sin(3x)} = \frac{\left(\lim_{x \to \frac{\pi}{2}} \frac{\cos(3x)}{\cos(5x)}\right)\sin\left(5\left(\lim_{x \to \frac{\pi}{2}} x\right)\right)}{\lim_{x \to \frac{\pi}{2}} \sin(3x)} \tag{76}$$

```
> Rule[identity]((76));
```

$$\lim_{x\to\frac{\pi}{2}} \frac{\sin(5x)\cos(3x)}{\cos(5x)\sin(3x)} = \frac{\lim_{x\to\frac{\pi}{2}} \frac{\cos(3x)}{\cos(5x)}}{\lim_{x\to\frac{\pi}{2}} \sin(3x)} \quad (77)$$

```
> Rule[sin]((77));
```

$$\lim_{x\to\frac{\pi}{2}} \frac{\sin(5x)\cos(3x)}{\cos(5x)\sin(3x)} = \frac{\lim_{x\to\frac{\pi}{2}} \frac{\cos(3x)}{\cos(5x)}}{\sin\left(\lim_{x\to\frac{\pi}{2}} 3x\right)} \quad (78)$$

```
> Rule[`c*`]((78));
```

$$\lim_{x\to\frac{\pi}{2}} \frac{\sin(5x)\cos(3x)}{\cos(5x)\sin(3x)} = \frac{\lim_{x\to\frac{\pi}{2}} \frac{\cos(3x)}{\cos(5x)}}{\sin\left(3\left(\lim_{x\to\frac{\pi}{2}} x\right)\right)} \quad (79)$$

```
> Rule[identity]((79));
```

$$\lim_{x\to\frac{\pi}{2}} \frac{\sin(5x)\cos(3x)}{\cos(5x)\sin(3x)} = -\left(\lim_{x\to\frac{\pi}{2}} \frac{\cos(3x)}{\cos(5x)}\right) \quad (80)$$

```
> Rule[`/`]((80));
This problem is an indeterminate form
```

$$\lim_{x\to\frac{\pi}{2}} \frac{\sin(5x)\cos(3x)}{\cos(5x)\sin(3x)} = -\left(\lim_{x\to\frac{\pi}{2}} \frac{\cos(3x)}{\cos(5x)}\right) \quad (81)$$

Évaluons séparément la limite du numérateur et la limite du dénominateur.

```
> Limit(cos(3*x),x=Pi/2)=limit(cos(3*x),x=Pi/2);
  Limit(cos(5*x),x=Pi/2)=limit(cos(5*x),x=Pi/2);
```

$$\lim_{x\to\frac{\pi}{2}} \cos(3x) = 0$$

$$\lim_{x\to\frac{\pi}{2}} \cos(5x) = 0 \quad (82)$$

On indentifie donc une indétermination du type $\frac{0}{0}$.

Afin d'appliquer la règle de L'Hospital, formons le quotient des dérivées.

```
> d:=x->-Diff(cos(3*x),x)/Diff(cos(5*x),x);
```

$$d := x \rightarrow -\frac{\frac{d}{dx}\cos(3x)}{\frac{d}{dx}\cos(5x)} \qquad (83)$$

Obtenons la limite du quotient des dérivées lorsque $x \rightarrow \frac{\pi}{2}$.

```
> Limit(-cos(3*x)/cos(5*x),x=Pi/2)=Limit(d(x),x=Pi/2);
```

$$\lim_{x \rightarrow \frac{\pi}{2}} \left(-\frac{\cos(3x)}{\cos(5x)}\right) = \lim_{x \rightarrow \frac{\pi}{2}} \left(-\frac{\frac{d}{dx}\cos(3x)}{\frac{d}{dx}\cos(5x)}\right) \qquad (84)$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d(x))),x=Pi/2);
```

$$= \lim_{x \rightarrow \frac{\pi}{2}} -\frac{3\sin(3x)}{5\sin(5x)} \qquad (85)$$

Appliquons les règles de calcul des limites.

```
> Rule[`/`]((85));
Creating problem #13 from the right hand side of this equation
```

$$\lim_{x \rightarrow \frac{\pi}{2}} -\frac{3\sin(3x)}{5\sin(5x)} = \frac{\lim_{x \rightarrow \frac{\pi}{2}} -\frac{3\sin(3x)}{5}}{\lim_{x \rightarrow \frac{\pi}{2}} \sin(5x)} \qquad (86)$$

```
> Rule[sin]((86));
```

$$\lim_{x \rightarrow \frac{\pi}{2}} -\frac{3\sin(3x)}{5\sin(5x)} = \frac{\lim_{x \rightarrow \frac{\pi}{2}} -\frac{3\sin(3x)}{5}}{\sin\left(\lim_{x \rightarrow \frac{\pi}{2}} 5x\right)} \qquad (87)$$

```
> Rule[`c*`]((87));
```

$$\lim_{x \rightarrow \frac{\pi}{2}} -\frac{3\sin(3x)}{5\sin(5x)} = \frac{\lim_{x \rightarrow \frac{\pi}{2}} -\frac{3\sin(3x)}{5}}{\sin\left(5\left(\lim_{x \rightarrow \frac{\pi}{2}} x\right)\right)} \qquad (88)$$

```
> Rule[identity]((88));
```

(89)

$$\lim_{x \to \frac{\pi}{2}} -\frac{3\sin(3x)}{5\sin(5x)} = \lim_{x \to \frac{\pi}{2}} -\frac{3\sin(3x)}{5} \qquad \textbf{(89)}$$

```
> Rule[`c*`]((89));
```

$$\lim_{x \to \frac{\pi}{2}} -\frac{3\sin(3x)}{5\sin(5x)} = -\frac{3\left(\lim_{x \to \frac{\pi}{2}} \sin(3x)\right)}{5} \qquad \textbf{(90)}$$

```
> Rule[sin]((90));
```

$$\lim_{x \to \frac{\pi}{2}} -\frac{3\sin(3x)}{5\sin(5x)} = -\frac{3\sin\left(\lim_{x \to \frac{\pi}{2}} 3x\right)}{5} \qquad \textbf{(91)}$$

```
> Rule[`c*`]((91));
```

$$\lim_{x \to \frac{\pi}{2}} -\frac{3\sin(3x)}{5\sin(5x)} = -\frac{3\sin\left(3\left(\lim_{x \to \frac{\pi}{2}} x\right)\right)}{5} \qquad \textbf{(92)}$$

```
> Rule[identity]((92));
```

$$\lim_{x \to \frac{\pi}{2}} -\frac{3\sin(3x)}{5\sin(5x)} = \frac{3}{5} \qquad \textbf{(93)}$$

L'indétermination est donc levée.

Réponse: $\lim_{x \to \frac{\pi}{2}} \frac{\tan(5x)}{\tan(3x)} = \frac{3}{5}$.

Explorons le tracé de la fonction f et celui du quotient des dérivées au voisinage de $x = \frac{\pi}{2}$ afin de constater que les deux tracés possèdent effectivement la même limite en $x = \frac{\pi}{2}$.

```
> plot([[x,f(x),x=Pi/3..2*Pi/3.1],[x,value(d(x)),x=1.2..1.9]],
  discont=true,color=[orange,navy],legend=["y=f(x)","y=d(x)"],view=
  [Pi/3..2*Pi/3.1,-0.5..1.5],tickmarks = [spacing((1/4)*Pi),
  default]);
```



Exemple 6

Soit la fonction *f* définie par $f(x) = \frac{1-\cos(x^3)}{x^6}$. Explorons le tracé de fonction *f* dans un voisinage de x = 0 .

```
> f:=x->(1-cos(x^3))/x^6;
  Tracé:=plot([x,f(x),x=-0.5..0.5],color=orange);
```

$$f := x \mapsto \frac{1-\cos(x^3)}{x^6}$$



Effectuons un zoom vers l'avant.

```
> display(Tracé,view=[-0.05..0.05,0.02..0.06]);
```



Le comportement erratique du tracé au voisinage de 0 est dû au nombre limité de chiffres utilisés par la macro-commande *plot* dans le calcul du quotient $\frac{1-\cos(x^3)}{x^6}$. C'est un aspect de la macro-commande *plot* où l'usager n'a aucun contrôle malheureusement. La solution qu'il faut adopter est de calculer un nombre suffisant de points puis de les relier dans un même graphique.

Pour le calcul des points, l'usager a évidemment le contrôle sur le nombre de décimales qui seront utilisés à cette fin. On prendra 50 chiffres après la virgule.

```
> L:= evalf[50]([seq([-0.5 + i/51,f(-.5 + i/51)],i=0..51)):
  pointplot(L,style=line,color=orange);
```



La fonction f n'est pas définie en $x = 0$. Si la limite $\lim_{x\to 0} \frac{1-\cos(x^3)}{x^6}$ existe, nous pourrons affirmer que la fonction f présente une discontinuité réductible en $x = 0$ puisque la fonction est toujours définie dans tout voisinage troué de $x = 0$.

```
> Limit(f(x),x=0)=limit(f(x),x=0);
```

$$\lim_{x \to 0} \frac{1-\cos(x^3)}{x^6} = \frac{1}{2} \qquad \textbf{(94)}$$

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites. Développons la limite du quotient comme le quotient des limites.

```
> Rule[`/`](Limit(f(x),x=0));
Creating problem #14

This problem is an indeterminate form
```

$$\lim_{x \to 0} \frac{1-\cos(x^3)}{x^6} = \lim_{x \to 0} \frac{1-\cos(x^3)}{x^6} \qquad \textbf{(95)}$$

Identifions la forme d'indétermination.

```
> Limit(numer(f(x)),x=0)=limit(numer(f(x)),x=0);
  Limit(denom(f(x)),x=0)=limit(denom(f(x)),x=0)
```

$$\lim_{x \to 0} \left(1-\cos(x^3)\right) = 0$$

$$\lim_{x \to 0} x^6 = 0 \quad \textbf{(96)}$$

On indentifie donc une indétermination du type $\frac{0}{0}$.

Afin d'appliquer la règle de L'Hospital, formons le quotient des dérivées.

```
> d:=x->Diff(numer(f(x)),x)/Diff(denom(f(x)),x);
```

$$d := x \to \frac{\frac{d}{dx}\, numer(f(x))}{\frac{d}{dx}\, denom(f(x))} \quad \textbf{(97)}$$

Appliquons le règle de L'Hospital. Développons la limite du quotient des dérivées lorsque $x \to 0$.

```
> Limit(f(x),x=0)=Limit(d(x),x=0);
```

$$\lim_{x \to 0} \frac{1 - \cos(x^3)}{x^6} = \lim_{x \to 0} \frac{\frac{d}{dx}\left(1 - \cos(x^3)\right)}{\frac{d}{dx}\left(x^6\right)} \quad \textbf{(98)}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d(x))),x=0);
```

$$= \lim_{x \to 0} \frac{\sin(x^3)}{2\,x^3} \quad \textbf{(99)}$$

Appliquons les règles de calcul des limites.

```
> Rule[`/`]((99));
Creating problem #15 from the right hand side of this equation
This problem is an indeterminate form
```

$$\lim_{x \to 0} \frac{\sin(x^3)}{2\,x^3} = \lim_{x \to 0} \frac{\sin(x^3)}{2\,x^3} \quad \textbf{(100)}$$

L'indétermination n'est donc pas encore levée. Évaluons séparément les limites du numérateur et du dénominateur.

```
> Limit(numer(value(d(x))),x=0)=limit(numer(value(d(x))),x=0);
  Limit(denom(value(d(x))),x=0)=limit(denom(value(d(x))),x=0);
```

$$\lim_{x \to 0} \sin(x^3) = 0$$

$$\lim_{x \to 0} 2\,x^3 = 0 \quad \textbf{(101)}$$

Nous avons donc encore identifié une indétermination du type $\frac{0}{0}$.

Appliquons de nouveau la règle de L'Hospital. Formons de nouveau le quotient des dérivées.

```
> d2:=x->Diff(numer(value(d(x))),x)/Diff(denom(value(d(x))),x);
```

(102)

$$d2 := x \rightarrow \frac{\frac{d}{dx}\, numer(value(d(x)))}{\frac{d}{dx}\, denom(value(d(x)))} \quad \textbf{(102)}$$

Appliquons le règle de L'Hospital.

```
> Limit(f(x),x=0)=Limit(d2(x),x=0);
```

$$\lim_{x \to 0} \frac{1-\cos(x^3)}{x^6} = \lim_{x \to 0} \frac{\frac{d}{dx}\, \sin(x^3)}{\frac{d}{dx}\, (2x^3)} \quad \textbf{(103)}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d2(x))),x=0);
```

$$= \lim_{x \to 0} \frac{\cos(x^3)}{2} \quad \textbf{(104)}$$

Appliquons les règles de calcul des limites.

```
> Rule[`c*`]((104));
Creating problem #16 from the right hand side of this equation
```

$$\lim_{x \to 0} \frac{\cos(x^3)}{2} = \frac{\left(\lim_{x \to 0} \cos(x^3)\right)}{2} \quad \textbf{(105)}$$

```
> Rule[cos]((105));
```

$$\lim_{x \to 0} \frac{\cos(x^3)}{2} = \frac{\cos\left(\lim_{x \to 0} x^3\right)}{2} \quad \textbf{(106)}$$

```
> Rule[`^`]((106));
```

$$\lim_{x \to 0} \frac{\cos(x^3)}{2} = \frac{\cos\left(\left(\lim_{x \to 0} x\right)^3\right)}{2} \quad \textbf{(107)}$$

```
> Rule[identity]((107));
```

$$\lim_{x \to 0} \frac{\cos(x^3)}{2} = \frac{1}{2} \quad \textbf{(108)}$$

L'indétermination est donc finalement levée après deux applications successives de la règle de l'Hospital.

Réponse: $\lim_{t \to 0} \frac{1-\cos(x^3)}{x^6} = \frac{1}{2}$.

Explorons le tracé de la fonction f et ceux des quotients des dérivées au voisinage de $x = 0$ afin de constater que tous les tracés possèdent effectivement la même limite en $x = 0$.

```
> L:= evalf[50]([seq]([-0.5 + i/51,f(-0.5 + i/51)],i=0..51)):
  L1:= evalf[50]([seq]([-0.5 + i/51,eval(value(d(x)),x=-0.5 + i/51)
  ],i=0..51)):
```

```
L2:= evalf[50]([seq]([-0.5 + i/51,eval(value(d2(x)),x=-0.5 + i/51)
],i=0..51)):
```

```
> C:=pointplot(L,style=line,color=orange):
  C1:=pointplot(L1,style=line,color=navy):
  C2:=pointplot(L2,style=line,color=khaki):
  display([C,C1,C2],legend=["y=f(x)","y=d(x)","y=d2(x)"]);
```



Indétermination de la forme $0\,(\pm\,\infty)$

Exemple 7

Soit la fonction f définie par $f(x) = \ln(x)\sqrt{x}$. La fonction f est définie seulement pour $x > 0$.

Explorons le tracé de la fonction f dans un voisinage à droite de $x = 0$.

```
> f:=x->ln(x)*sqrt(x);
```

$$f := x \mapsto \ln(x)\cdot\sqrt{x} \tag{109}$$

```
> Tracé:=plot([x,f(x),x=0..1],color=orange,view=[-0.2..1,-1..0.5]):
  Tracé;
```



Les voisins à la droite de $x = 0$ semblent avoir des images s'approchant d'un certain nombre. Zoomons vers l'avant dans ce voisinage pour mieux estimer ce nombre.

```
> display(Tracé,view=[-0.01..0.04,-0.5..0.05]);
```



Il aurait fallu, au départ, réaliser le tracé avec un nombre minimum de points supérieur au nombre par défaut (200) afin de mieux être en mesure de faire cette estimation.

```
> Tracé:=plot([x,f(x),x=0..0.05],color=orange,numpoints=400,view=
  [-0.01..0.05,-1..0.2]):
  Tracé;
```



Calculons directement la limite $\lim_{x\to 0+} \ln(x)\sqrt{x}$ avec la macro-commande *limit*.

```
> Limit(f(x),x=0,right)=limit(f(x),x=0,right);
```

$$\lim_{x\to 0+} \ln(x)\sqrt{x} = 0 \tag{110}$$

Cela confirme donc que la fonction *f* présente une discontinuité à droite réductible en $x = 0$.

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites.

```
> Rule[`*`](Limit(f(x),x=0,right));
Creating problem #17

This problem is an indeterminate form
```

$$\lim_{x\to 0+} \ln(x)\sqrt{x} = \lim_{x\to 0+} \ln(x)\sqrt{x} \tag{111}$$

Selon les modèles utilisés dans la fonction *f*, l'indétermination est du type $(-\infty)\,0^{+}$.

Transformons cette forme d'infétermination en la forme $\frac{-\infty}{\infty}$

```
> Rule[change,x=1/t]((111));
Applying substitution x = 1/t, t = 1/x
```

$$\lim_{x \to 0+} \ln(x)\sqrt{x} = \lim_{t \to \infty} \frac{\ln\left(\frac{1}{t}\right)}{\sqrt{t}} \tag{112}$$

```
> Rule[`/`]((112));
This problem is an indeterminate form
```

$$\lim_{x \to 0+} \ln(x)\sqrt{x} = \lim_{t \to \infty} \frac{\ln\left(\frac{1}{t}\right)}{\sqrt{t}} \tag{113}$$

Ne soyons donc pas surpris !!

```
> Limit(ln(1/t),t=infinity)=limit(ln(1/t),t=infinity);
  Limit(sqrt(t),t=infinity)=limit(sqrt(t),t=infinity);
```

$$\lim_{t \to \infty} \ln\left(\frac{1}{t}\right) = -\infty$$

$$\lim_{t \to \infty} \sqrt{t} = \infty \tag{114}$$

Il s'agit d'une indétemination de type $\frac{-\infty}{\infty}$.

Afin d'appliquer la règle de L'Hospital, formons le quotient des dérivées.

```
> d:=t->Diff(ln(1/t),t)/Diff(sqrt(t),t);
```

$$d := t \to \frac{\frac{d}{dt}\ln\left(\frac{1}{t}\right)}{\frac{d}{dt}\sqrt{t}} \tag{115}$$

Posons ensuite la limite à droite en $t = \infty$ du quotient des dérivées.

```
> Limit(f(x),x=0,right)=Limit(d(t),t=infinity);
```

$$\lim_{x \to 0+} \ln(x)\sqrt{x} = \lim_{t \to \infty} \frac{\frac{d}{dt}\ln\left(\frac{1}{t}\right)}{\frac{d}{dt}\left(\sqrt{t}\right)} \tag{116}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d(t))),t=infinity);
```

$$= \lim_{t \to \infty} -\frac{2}{\sqrt{t}} \tag{117}$$

Appliquons les règles du calcul des limites.

```
> Rule[`c*`]((117));
Creating problem #18 from the right hand side of this equation
```

$$\lim_{t \to \infty} -\frac{2}{\sqrt{t}} = -2\left(\lim_{t \to \infty} \frac{1}{\sqrt{t}}\right) \tag{118}$$

```
> Rule[`/`]((118));
```

$$\lim_{t\to\infty} -\frac{2}{\sqrt{t}} = -\frac{2\left(\lim_{t\to\infty} 1\right)}{\lim_{t\to\infty}\sqrt{t}} \quad \mathbf{(119)}$$

```
> Rule[constant]((119));
```

$$\lim_{t\to\infty} -\frac{2}{\sqrt{t}} = -\frac{2}{\lim_{t\to\infty}\sqrt{t}} \quad \mathbf{(120)}$$

```
> Rule[`^`]((120));
```

$$\lim_{t\to\infty} -\frac{2}{\sqrt{t}} = -\frac{2}{\sqrt{\lim_{t\to\infty} t}} \quad \mathbf{(121)}$$

```
> Rule[identity]((121));
```

$$\lim_{t\to\infty} -\frac{2}{\sqrt{t}} = 0 \quad \mathbf{(122)}$$

L'indétermination est donc levée.

Réponse: $\lim_{x\to 0+} \ln(x)\sqrt{x} = 0$.

Explorons les tracé de $t \mapsto \frac{\ln\left(\frac{1}{t}\right)}{\sqrt{t}}$ et de $t \mapsto \frac{Diff\left(\ln\left(\frac{1}{t}\right), t\right)}{Diff\left(\sqrt{t}, t\right)}$ découlant du changement de variables $x = \frac{1}{t}$. lorsque $t \to \infty$, $x \to 0^+$

```
> plot([[t,ln(1/t)/sqrt(t),t=0..500],[t,diff(ln(1/t),t)/diff(sqrt
  (t),t),t=0..500]],color=[orange,navy],legend=["y=f(t)","y=d(t)"],
  view=[-0.2..200,-1..0.5]);
```



Exemple 8

Soit la fonction f définie par $f(x) = x\left(1 - \cos\left(\frac{1}{x}\right)\right)$. Explorons le comportement de la fonction lorsque

$x \rightarrow \infty$.

```
> f:=x->x*(1-cos(1/x));
```

$$f := x \mapsto x \cdot \left(1 - \cos\left(\frac{1}{x}\right)\right) \quad \mathbf{(123)}$$

```
> plot([x,f(x),x=4..100],color=orange,view=[0..100,-0.2..0.2]);
```



Il semble que les images possèdent un comportement asymptotique. Évaluons la limite à l'infini directement avec la macro-commande *limit*.

```
> Limit(f(x),x=infinity)=limit(f(x),x=infinity);
```

$$\lim_{x \rightarrow \infty} x \left(1 - \cos\left(\frac{1}{x}\right)\right) = 0 \quad \mathbf{(124)}$$

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites.

```
> Rule[`*`](Limit(f(x),x=infinity));
Creating problem #19

This problem is an indeterminate form
```

$$\lim_{x \rightarrow \infty} x \left(1 - \cos\left(\frac{1}{x}\right)\right) = \lim_{x \rightarrow \infty} x \left(1 - \cos\left(\frac{1}{x}\right)\right) \quad \mathbf{(125)}$$

On reconnaît une indétermination de la forme $\infty \cdot 0$.

```
> Rule[change,t=1/x]((125));
Applying substitution x = 1/t, t = 1/x
```

$$\lim_{x \rightarrow \infty} x \left(1 - \cos\left(\frac{1}{x}\right)\right) = \lim_{t \rightarrow 0+} \left(\frac{1}{t} - \frac{\cos(t)}{t}\right) \quad \mathbf{(126)}$$

```
> normal((126));
```

$$\lim_{x \rightarrow \infty} x \left(1 - \cos\left(\frac{1}{x}\right)\right) = \lim_{t \rightarrow 0+} \left(-\frac{-1 + \cos(t)}{t}\right) \quad \mathbf{(127)}$$

```
> Rule[`c*`]((127));
```

$$\lim_{x \rightarrow \infty} x \left(1 - \cos\left(\frac{1}{x}\right)\right) = -\left(\lim_{t \rightarrow 0+} \frac{-1 + \cos(t)}{t}\right) \quad \mathbf{(128)}$$

```
> Rule[`/`]((128));
This problem is an indeterminate form
```

$$\lim_{x \to \infty} x \left(1 - \cos\left(\frac{1}{x}\right)\right) = -\left(\lim_{t \to 0+} \frac{-1 + \cos(t)}{t}\right) \tag{129}$$

On a affaire à une indétermination de la forme $\frac{0}{0}$. Appliquons la règle de l'Hospital.

Afin d'appliquer la règle de L'Hospital, formons le quotient des dérivées.

```
> d:=t->Diff(-1+cos(t),t)/Diff(t,t);
```

$$d := t \to \frac{\frac{d}{dt}(-1 + \cos(t))}{\frac{d}{dt} t} \tag{130}$$

Appliquons le règle de L'Hospital.

```
> Limit(f(x),x=infinity)=Limit(d(t),t=0,right);
```

$$\lim_{x \to \infty} x \left(1 - \cos\left(\frac{1}{x}\right)\right) = \lim_{t \to 0+} \frac{\frac{d}{dt}(-1 + \cos(t))}{\frac{d}{dt} t} \tag{131}$$

```
> ``=Limit(normal(value(d(t))),t=0,right);
```

$$= \lim_{t \to 0+} (-\sin(t)) \tag{132}$$

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites.

```
> Rule[`c*`]((132));
Creating problem #20 from the right hand side of this equation
```

$$\lim_{t \to 0+} (-\sin(t)) = -\left(\lim_{t \to 0+} \sin(t)\right) \tag{133}$$

```
> Rule[sin]((133));
```

$$\lim_{t \to 0+} (-\sin(t)) = 0 \tag{134}$$

L'indétermination est donc levée.

Réponse: $\lim_{x \to \infty} x \left(1 - \cos\left(\frac{1}{x}\right)\right) = 0$.

Indétermination de la forme $\infty - \infty$

Exemple 9

Soit la fonction *f* définie par $f(x) = \frac{1}{\arctan(x)} - \frac{1}{x}$. Cette fonction n'est pas définie en $x = 0$. Explorons le comportement de la fonction *f* dans un voisinage de $x = 0$.

```
> f:=x->1/arctan(x)-1/x;
```

$$f := x \mapsto \frac{1}{\arctan(x)} - \frac{1}{x} \tag{135}$$

```
> plot([x,f(x),x=-Pi..Pi],color=orange,view=[-Pi..Pi,-1..1],
  tickmarks = [spacing((1/4)*Pi), default]);
```



Il semble donc que la fonction *f* possède en $x = 0$ une discontinuité réductible. Confirmons cette affirmation en évaluant la limite avec la macro-commande *limit*.

```
> Limit(f(x),x=0)=limit(f(x),x=0);
```

$$\lim_{x \to 0} \left(\frac{1}{\arctan(x)} - \frac{1}{x} \right) = 0 \tag{136}$$

Il se présente donc en $x = 0$, une discontinuité réductible.

Calculons autrement la limite $\lim_{x\to 0} \left(\frac{1}{\arctan(x)} - \frac{1}{x} \right)$ en appliquant *pas-à-pas* les règles de calcul des limites.

```
> Rule[`-`](Limit(f(x),x=0));
Creating problem #21

This problem is an indeterminate form
```

$$\lim_{x \to 0} \left(\frac{1}{\arctan(x)} - \frac{1}{x} \right) = \lim_{x \to 0} \left(\frac{1}{\arctan(x)} - \frac{1}{x} \right) \tag{137}$$

```
> Limit(1/arctan(x),x=0,right)=limit(1/arctan(x),x=0,right);
  Limit(1/x,x=0,right)=limit(1/x,x=0,right);
```

$$\lim_{x \to 0+} \frac{1}{\arctan(x)} = \infty$$

$$\lim_{x \to 0+} \frac{1}{x} = \infty \tag{138}$$

```
> Limit(1/arctan(x),x=0,left)=limit(1/arctan(x),x=0,left);
  Limit(1/x,x=0,left)=limit(1/x,x=0,left);
```

$$\lim_{x \to 0-} \frac{1}{\arctan(x)} = -\infty$$

$$\lim_{x \to 0-} \frac{1}{x} = -\infty \qquad (139)$$

On reconnaît une indétermination de la forme $\infty - \infty (-\infty + \infty)$

```
> normal((137));
```

$$\lim_{x \to 0} \left(\frac{1}{\arctan(x)} - \frac{1}{x} \right) = \lim_{x \to 0} \left(-\frac{-x + \arctan(x)}{\arctan(x)\, x} \right) \qquad (140)$$

```
> Rule[`c*`]((140));
```

$$\lim_{x \to 0} \left(\frac{1}{\arctan(x)} - \frac{1}{x} \right) = -\left(\lim_{x \to 0} \frac{-x + \arctan(x)}{\arctan(x)\, x} \right) \qquad (141)$$

```
> Rule[`/`]((141));
This problem is an indeterminate form
```

$$\lim_{x \to 0} \left(\frac{1}{\arctan(x)} - \frac{1}{x} \right) = -\left(\lim_{x \to 0} \frac{-x + \arctan(x)}{\arctan(x)\, x} \right) \qquad (142)$$

```
> Limit(-x+arctan(x),x=0)=limit(-x+arctan(x),x=0);
  Limit(arctan(x)*x,x=0)=limit(arctan(x)*x,x=0);
```

$$\lim_{x \to 0} (-x + \arctan(x)) = 0$$

$$\lim_{x \to 0} \arctan(x)\, x = 0 \qquad (143)$$

Il s'agit d'une indétermination du type $\frac{0}{0}$.

Afin d'appliquer la règle de L'Hospital, créons les fonctions $f1$ et $f2$ définie par $f1(x) = -x + \arctan(x)$ et $f2(x) = \arctan(x)\, x$.

```
> f1:=x->-x+arctan(x);
  f2:=x->arctan(x)*x;
```

$$f1 := x \mapsto -x + \arctan(x)$$

$$f2 := x \mapsto \arctan(x) \cdot x \qquad (144)$$

Afin d'appliquer la règle de L'Hospital, formons le quotient des dérivées.

```
> d:=x->Diff(f1(x),x)/Diff(f2(x),x);
```

(145)

$$d := x \mapsto \frac{\frac{d}{dx} f1(x)}{\frac{d}{dx} f2(x)} \tag{145}$$

Appliquons le règle de L'Hospital. Développons la limite en $x = 0$ du quotient des dérivées.

```
> Limit(f(x),x=0,right)=Limit(d(x),x=0,right);
```

$$\lim_{x \to 0+} \left(\frac{1}{\arctan(x)} - \frac{1}{x} \right) = \lim_{x \to 0+} \frac{\frac{d}{dx} (-x + \arctan(x))}{\frac{d}{dx} (\arctan(x)\, x)} \tag{146}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d(x))),x=0,right);
```

$$= \lim_{x \to 0+} \left(-\frac{x^2}{\arctan(x)\, x^2 + \arctan(x) + x} \right) \tag{147}$$

Appliquons les règles de calcul des limites.

```
> Rule[`/`]((147));
Creating problem #22 from the right hand side of this equation
This problem is an indeterminate form
```

$$\lim_{x \to 0+} \left(-\frac{x^2}{\arctan(x)\, x^2 + \arctan(x) + x} \right) = \lim_{x \to 0+} \left(-\frac{x^2}{\arctan(x)\, x^2 + \arctan(x) + x} \right) \tag{148}$$

À nouveau, on a affaire à une indétermination du type $\frac{0}{0}$.

Nous pouvons donc encore appliquer la règle de L'Hospital. Formons alors un nouveau quotient des dérivées.

```
> d2:=x->Diff(numer(normal(value(d(x)))),x)/Diff(denom(normal(value
  (d(x)))),x);
```

$$d2 := x \rightarrow \frac{\frac{d}{dx} numer(normal(value(d(x))))}{\frac{d}{dx} denom(normal(value(d(x))))} \tag{149}$$

Appliquons le règle de L'Hospital. Développons la limite en $x = 0$ du quotient des dérivées.

```
> Limit(f(x),x=0,right)=Limit(d2(x),x=0,right);
```

$$\lim_{x \to 0+} \left(\frac{1}{\arctan(x)} - \frac{1}{x} \right) = \lim_{x \to 0+} \frac{\frac{d}{dx} (-x^2)}{\frac{d}{dx} \left(\arctan(x)\, x^2 + \arctan(x) + x\right)} \tag{150}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d2(x))),x=0,right);
```

$$= \lim_{x \to 0+} \left(-\frac{x}{\arctan(x)\, x + 1} \right) \tag{151}$$

Appliquons les règles de calcul des limites.

```
> Rule[`/`]((151));
Creating problem #23 from the right hand side of this equation
```

$$\lim_{x \to 0+} \left(-\frac{x}{\arctan(x)\,x+1} \right) = \frac{\lim_{x \to 0+} (-x)}{\lim_{x \to 0+} (\arctan(x)\,x+1)} \quad (152)$$

```
> Rule[`c*`]((152));
```

$$\lim_{x \to 0+} \left(-\frac{x}{\arctan(x)\,x+1} \right) = -\frac{\lim_{x \to 0+} x}{\lim_{x \to 0+} (\arctan(x)\,x+1)} \quad (153)$$

```
> Rule[identity]((153));
```

$$\lim_{x \to 0+} \left(-\frac{x}{\arctan(x)\,x+1} \right) = 0 \quad (154)$$

L'indétermination est donc levée.

Réponse: $\lim_{x \to 0+} \left(\frac{1}{\arctan(x)} - \frac{1}{x} \right) = 0.$

Explorons le tracé de la fonction f et ceux des quotients des dérivées au voisinage de $x = 0$ afin de constater que tous les tracés possèdent effectivement la même limite en $x = 0$.

```
> plot([[x,f(x),x=0..Pi/6],[x,value(d(x)),x=0..Pi/6],[x,value(d2(x)
  ),x=0..Pi/6]],color=[orange,navy,khaki],legend=["y=f(x)","y=d(x)",
  "y=d2(x)"],view=[-0.1..Pi/6,-0.12..0.2]);
```



Exemple 10

Soit la fonction f définie par $f(x) = \frac{1}{x-2} + \frac{4}{4-x^2}$.

```
> f:=x->1/(x-2)+4/(4-x^2);
```

$$f := x \mapsto \frac{1}{x-2} + \frac{4}{4-x^2} \quad (155)$$

```
> plot([x,f(x),x=-3..3],discont=true,color=orange,view=[-3..3,-4..4]
  );
```



Il n'y a aucune discontinuité graphiquement apparente en $x = 2$. Puisque la fonction f n'est pas définie en $x = 2$ (ni en $x = -2$), il semble donc qu'il se présente $x = 2$ une **discontinuité réductible**. De plus, on remarque que les voisins de droite comme ceux de gauche de $x = 2$ ont des images voisines de 0,25 (Il suffit de cliquer au bon endroit sur le graphique après avoir obtenu un agrandissement à 400%. On obtient ce zoom avant en cliquant d'abord sur le graphique afin de le sélectionner puis en pressant Ctrl-6. Ou bien, sélectionnez la rubrique appropriée du sous-menu «Zoom Factor» du menu «View». Ctrl-2 vous permettra de rétablir l'affichage à 100%.).

Évaluons directement la limite avec la macro-commande *limit*.

```
> Limit(f(x),x=2)=limit(f(x),x=2);
```

$$\lim_{x \to 2} \left(\frac{1}{x-2} + \frac{4}{-x^2+4} \right) = \frac{1}{4} \tag{156}$$

La fonction f possède effectivement une discontinuité réductible en $x = 0$.

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites.

```
> Rule[`+`](Limit(f(x),x=2));
Creating problem #24

This problem is an indeterminate form
```

$$\lim_{x \to 2} \left(\frac{1}{x-2} + \frac{4}{-x^2+4} \right) = \lim_{x \to 2} \left(\frac{1}{x-2} + \frac{4}{-x^2+4} \right) \tag{157}$$

Fallait s'y attendre, On a une indétermination de la forme $\infty - \infty$.

Récrivons la fonction sur un même dénominateur.

```
> normal((157));
```

$$\lim_{x \to 2} \left(\frac{1}{x-2} + \frac{4}{-x^2+4} \right) = \lim_{x \to 2} \frac{1}{x+2} \tag{158}$$

```
> Rule[`/`]((158));
```

$$\lim_{x \to 2} \left(\frac{1}{x-2} + \frac{4}{-x^2+4} \right) = \frac{\lim_{x \to 2} 1}{\lim_{x \to 2} (x+2)} \tag{159}$$

```
> Rule[constant]((159));
```

$$\lim_{x\to 2}\left(\frac{1}{x-2}+\frac{4}{-x^2+4}\right)=\frac{1}{\lim_{x\to 2}(x+2)} \quad (160)$$

```
> Rule[`+`]((160));
```

$$\lim_{x\to 2}\left(\frac{1}{x-2}+\frac{4}{-x^2+4}\right)=\frac{1}{\left(\lim_{x\to 2}x\right)+\left(\lim_{x\to 2}2\right)} \quad (161)$$

```
> Rule[identity]((161));
```

$$\lim_{x\to 2}\left(\frac{1}{x-2}+\frac{4}{-x^2+4}\right)=\frac{1}{2+\left(\lim_{x\to 2}2\right)} \quad (162)$$

```
> Rule[constant]((162));
```

$$\lim_{x\to 2}\left(\frac{1}{x-2}+\frac{4}{-x^2+4}\right)=\frac{1}{4} \quad (163)$$

L'indétermination est donc levée.

Réponse: $\lim_{x\to 2}\left(\frac{1}{x-2}+\frac{4}{4-x^2}\right)=\frac{1}{4}$.

Indétermination de la forme 0^0

Exemple 11

Soit la fonction f définie par $f(x) = x^{\sin(x)}$. Le domaine de la fonction f étant l'intervalle $]0, \infty[$, explorons le tracé de la fonction f avec un voisinage à la droite de 0.

```
> f:=x->x^sin(x);
```

$$f := x \mapsto x^{\sin(x)} \quad (164)$$

```
> plot([x,f(x),x=0..2],color=orange,view=[-0.5..2,-1..2]);
```



Les voisins de droite en $x = 0$ semblent avoir des images s'approchant de $y = 1$. Puisque la fonction f n'est pas définie en $x = 0$ et compte tenu du tracé de la fonction f, il semble donc se présenter en $x = 0$ une discontinuité réductible à droite.

Évaluons directement la limite pour le montrer.

```
> Limit(f(x),x=0,right)=limit(f(x),x=0,right);
```

$$\lim_{x \to 0+} x^{\sin(x)} = 1 \tag{165}$$

Il se présente donc, en $x = 0$, une discontinuité à droite réductible.

Calculons autrement cette limite en appliquant *pas-à-pas* les règles de calcul des limites.

```
> Rule[`^`](Limit(f(x),x=0,right));
Creating problem #25

This problem is an indeterminate form
```

$$\lim_{x \to 0+} x^{\sin(x)} = \lim_{x \to 0+} x^{\sin(x)} \tag{166}$$

On a une indétermination de la forme 0^0.

```
> Rule[rewrite,x^sin(x)=exp(sin(x)*ln(x))]((166));
```

$$\lim_{x \to 0+} x^{\sin(x)} = \lim_{x \to 0+} e^{\sin(x)\ln(x)} \tag{167}$$

Effectuons un passage à la limite.

```
> Rule[exp]((167));
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{\lim_{x \to 0+} \sin(x)\ln(x)} \tag{168}$$

```
> Rule[`*`]((168));
This problem is an indeterminate form
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{\lim_{x \to 0+} \sin(x)\ln(x)} \tag{169}$$

La limite en position d'exposant est une indétermination de la forme $0 \cdot (-\infty)$.

Réécrivons cette limite en une indétermination de la forme $\frac{-\infty}{+\infty}$: $e^{\lim_{x \to 0+} \frac{\ln(x)}{\frac{1}{\sin(x)}}}$

```
> Rule[rewrite, sin(x)=1/(1/sin(x))]((169));
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{\lim_{x \to 0+} \sin(x)\ln(x)} \tag{170}$$

Le mécanisme de la simplification automatique ne le permet pas évidemment.

```
> Rule[rewrite, sin(x)=1/csc(x)]((169));
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{\lim_{x \to 0+} \frac{\ln(x)}{\csc(x)}} \quad \textbf{(171)}$$

```
> Rule[rewrite,ln(x)=diff(ln(x),x), csc(x)=diff(csc(x),x)]((171));
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{\lim_{x \to 0+} \left(-\frac{1}{\csc(x)\cot(x)\, x}\right)} \quad \textbf{(172)}$$

```
> convert((172),sincos);
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{\lim_{x \to 0+} \left(-\frac{\sin(x)^2}{\cos(x)\, x}\right)} \quad \textbf{(173)}$$

```
> Rule[`c*`]((173));
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{-\left(\lim_{x \to 0+} \frac{\sin(x)^2}{\cos(x)\, x}\right)} \quad \textbf{(174)}$$

```
> Rule[`/`]((174));
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{-\frac{\lim_{x \to 0+} \frac{\sin(x)^2}{x}}{\lim_{x \to 0+} \cos(x)}} \quad \textbf{(175)}$$

Rappel: Les règles *sum*, *difference*, *product*, *quotient* et *devidebyzero* ne sciendent pas les cas d'indéterminations.

```
> Rule[cos]((175));
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{-\left(\lim_{x \to 0+} \frac{\sin(x)^2}{x}\right)} \quad \textbf{(176)}$$

Nous avons encore une indtermination du type $\frac{0}{0}$.

```
> Rule[rewrite,sin(x)^2=diff(sin(x)^2,x),1/x=1/diff(x,x)]((176));
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{-\left(\lim_{x \to 0+} 2\sin(x)\cos(x)\right)} \quad \textbf{(177)}$$

```
> Rule[`c*`]((177));
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{-2\left(\lim_{x \to 0+} \sin(x)\cos(x)\right)} \quad \textbf{(178)}$$

```
> Rule[`*`]((178));
```

$$\lim_{x \to 0+} x^{\sin(x)} = e^{-2 \left(\lim_{x \to 0+} \sin(x)\right) \left(\lim_{x \to 0+} \cos(x)\right)} \quad (179)$$

```
> Rule[sin]((179));
```

$$\lim_{x \to 0+} x^{\sin(x)} = 1 \quad (180)$$

L'indétermination est donc levée.

Réponse: $\lim_{x\to 0+} x^{\sin(x)} = 1$.

Inefficacité de la règle de L'Hospital

Exemple 12

Soit la fonction f définie par $f(x) = \dfrac{\sqrt{x^2+1}}{x}$ dont le domaine est $\mathbb{R} \setminus \{0\}$. Explorons le comportement de la fonction f lorsque $x \to -\infty$.

```
> f:=x->sqrt(x^2+1)/x;
```

$$f := x \mapsto \frac{\sqrt{x^2+1}}{x} \quad (181)$$

```
> plot([x,f(x),x=-25..2],color=orange,view=[-25..2,-2..0]);
```



Le tracé de la fonction f semble présenter un comportement aymptotique horizontal lorsque $x \to -\infty$. Évaluons la limite à $-\infty$ pour en être convaincu.

```
> Limit(f(x),x=-infinity)=limit(f(x),x=-infinity);
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2+1}}{x} = -1 \quad (182)$$

Mettons en évidence que cette limite est une indétermination de type $\dfrac{\infty}{-\infty}$.

```
> Limit(numer(f(x)),x=infinity)=limit(numer(f(x)),x=-infinity);
  Limit(denom(f(x)),x=infinity)=limit(denom(f(x)),x=-infinity);
```

$$\lim_{x \to \infty} \sqrt{x^2+1} = \infty$$

$$\lim_{x \to \infty} x = -\infty \quad \textbf{(183)}$$

On a effectivement une indétermination de la forme $\frac{\infty}{-\infty}$.

Une fois n'est pas coutume, appliquons la règle *lhopital* de la bibliothèque *Student*[*Calculus1*] après avoir identifié une indétermination.

```
> Rule[`/`](Limit(f(x),x=-infinity));
Creating problem #26

This problem is an indeterminate form
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2+1}}{x} = \lim_{x \to -\infty} \frac{\sqrt{x^2+1}}{x} \quad \textbf{(184)}$$

La règle *quotient* a mis en évidence une indétermination… appliquons alors la règle *lhopital*.

```
> Rule[lhopital,sqrt(x^2+1)]((184));
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2+1}}{x} = \lim_{x \to -\infty} \frac{x}{\sqrt{x^2+1}} \quad \textbf{(185)}$$

Voyons ce que donnera la règle *quotient* avec cette nouvelle limte.

```
> Rule[`/`]((185));
This problem is an indeterminate form
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2+1}}{x} = \lim_{x \to -\infty} \frac{x}{\sqrt{x^2+1}} \quad \textbf{(186)}$$

Nous avons encore une indétermination. Appliquons de nouveau *lhopital*.

```
> Rule[lhopital,x]((186));
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2+1}}{x} = \lim_{x \to -\infty} \frac{\sqrt{x^2+1}}{x} \quad \textbf{(187)}$$

Nous voici revenu à la case départ. La règle *lhopital* s'avère donc inoppérante.

Pour lever l'indétermination, il va falloir réaliser une développement différent.

Effectuons une pseudo mise en évidence de x^2 du radicand x^2+1. Nous allons utiliser la macro-commande quo pour effectuer la division polynomiale.

```
> Radicand:=x^2*(quo(x^2+1,x^2,x,'reste')+reste/x^2);
```

$$Radicand := x^2 \left(1 + \frac{1}{x^2}\right) \quad \textbf{(188)}$$

Créons une nouvelle expression pour la fonction *f*.

```
> Nouveau_numérateur:=sqrt(Radicand);
```

$$Nouveau\_numérateur := \sqrt{x^2\left(1+\frac{1}{x^2}\right)} \quad \textbf{(189)}$$

```
> Nouvelle_formulation_f:=x->Nouveau_numérateur/x;
```

$$Nouvelle\_formulation\_f := x \mapsto \frac{Nouveau\_numérateur}{x} \quad \textbf{(190)}$$

```
> Limit(f(x),x=-infinity)=Limit(Nouvelle_formulation_f(x),x=-infinity);
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2+1}}{x} = \lim_{x \to -\infty} \frac{\sqrt{x^2\left(1+\frac{1}{x^2}\right)}}{x} \quad \textbf{(191)}$$

Forçons la simplification avec un changement de variables.

```
> Rule[change,x=-1/t]((191));
Creating problem #27 from the right hand side of this equation
Applying substitution x = -1/t, t = -1/x
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2\left(1+\frac{1}{x^2}\right)}}{x} = \lim_{t \to 0+} \left(-\sqrt{t^2+1}\right) \quad \textbf{(192)}$$

```
> Rule[`c*`]((192));
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2\left(1+\frac{1}{x^2}\right)}}{x} = -\left(\lim_{t \to 0+} \sqrt{t^2+1}\right) \quad \textbf{(193)}$$

Effwctuons un passage à la limite.

```
> Rule[`^`]((193));
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2\left(1+\frac{1}{x^2}\right)}}{x} = -\sqrt{\lim_{t \to 0+} \left(t^2+1\right)} \quad \textbf{(194)}$$

```
> Rule[`+`]((194));
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2\left(1+\frac{1}{x^2}\right)}}{x} = -\sqrt{\left(\lim_{t \to 0+} t^2\right) + \left(\lim_{t \to 0+} 1\right)} \quad \textbf{(195)}$$

```
> Rule[constant]((195));
```

(196)

$$\lim_{x \to -\infty} \frac{\sqrt{x^2\left(1+\frac{1}{x^2}\right)}}{x} = -\sqrt{\left(\lim_{t \to 0+} t^2\right)+1} \quad \textbf{(196)}$$

```
> Rule[`^`]((196));
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2\left(1+\frac{1}{x^2}\right)}}{x} = -\sqrt{\left(\lim_{t \to 0+} t\right)^2+1} \quad \textbf{(197)}$$

```
> Rule[identity]((197));
```

$$\lim_{x \to -\infty} \frac{\sqrt{x^2\left(1+\frac{1}{x^2}\right)}}{x} = -1 \quad \textbf{(198)}$$

L'indétermination est donc levée.

Réponse: $\lim_{x \to -\infty} \frac{\sqrt{x^2+1}}{x} = -1$.

Exemple 13

Soit la fonction *f* définie par $f(x) = \frac{3x^2+\sin(2x)}{x^2+\cos(3x)}$. Explorons le comportement de la fonction *f* lorsque $x \to \infty$.

```
> f:=x->(3*x^2+sin(2*x))/(x^2+cos(3*x));
```

$$f := x \mapsto \frac{3 \cdot x^2+\sin(2 \cdot x)}{x^2+\cos(3 \cdot x)} \quad \textbf{(199)}$$

```
> plot([[x,f(x),x=10..50],[x,3,x=10..50]],numpoints=120,color=
  [orange,khaki],linestyle=[1,2]);
```



Lorsque $x \to \infty$, les images $f(x)$ semblent se rapprocher de 3 par oscillation. Confirmons cette observation graphique par le calcul de la limite $\lim_{x \to \infty} f(x)$.

```
> Limit(f(x),x=infinity)=limit(f(x),x=infinity);
```

$$\lim_{x \to \infty} \frac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} = 3 \tag{200}$$

Aussi, lorsque $x \to \infty$, $\dfrac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} \approx \dfrac{3x^2}{x^2} = 3$, étant donné que les apports numériques de $\sin(2x)$ au numérateur et de $\cos(3x)$ au dénominateur deviennent respectivement de plus en plus négligeable au calcul de la fraction lorsque $x \to \infty$.

Comme dans l'exemple précédent, on utilisera la règle *lhopital*.

```
> Rule[`/`](Limit(f(x),x=infinity));
Creating problem #28

This problem is an indeterminate form
```

$$\lim_{x \to \infty} \frac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} = \lim_{x \to \infty} \frac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} \tag{201}$$

La règle *quotient* a mis en évidence une indétermination… appliquons alors la règle *lhopital*.

```
> Rule[lhopital,3*x^2+sin(2*x)]((201));
```

$$\lim_{x \to \infty} \frac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} = \lim_{x \to \infty} -\frac{2\,(3x + \cos(2x))}{-2x + 3\sin(3x)} \tag{202}$$

```
> Rule[`c*`]((202));
```

$$\lim_{x \to \infty} \frac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} = -2\left(\lim_{x \to \infty} \frac{3x + \cos(2x)}{-2x + 3\sin(3x)}\right) \tag{203}$$

```
> Rule[`/`]((203));
This problem is an indeterminate form
```

$$\lim_{x \to \infty} \frac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} = -2\left(\lim_{x \to \infty} \frac{3x + \cos(2x)}{-2x + 3\sin(3x)}\right) \tag{204}$$

La règle *quotient* a encore mis en évidence une indétermination… appliquons alors la règle *lhopital*.

```
> Rule[lhopital,(3*x+cos(2*x))]((204));
```

$$\lim_{x \to \infty} \frac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} = -2\left(\lim_{x \to \infty}\left(-\frac{-3 + 2\sin(2x)}{-2 + 9\cos(3x)}\right)\right) \tag{205}$$

```
> Rule[`c*`]((205));
```

$$\lim_{x \to \infty} \frac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} = 2\left(\lim_{x \to \infty} \frac{-3 + 2\sin(2x)}{-2 + 9\cos(3x)}\right) \tag{206}$$

```
> Rule[`/`]((206));
Rule [quotient] does not apply
```

$$\lim_{x\to\infty} \frac{3x^2+\sin(2x)}{x^2+\cos(3x)} = 2\left(\lim_{x\to\infty} \frac{-3+2\sin(2x)}{-2+9\cos(3x)}\right) \quad \mathbf{(207)}$$

Oups! La règle *quotient* ne nous permet plus de poursuivre avec les règles de calculs de limites (et ce n'est pas une forme d'indétermination).

Nous allons reprendre ce calcul de limite dans l'esprit de ce document après avoir identifié la forme d'indétermination.

```
>  Rule[`/`](Limit(f(x),x=infinity));
Creating problem #29

This problem is an indeterminate form
```

$$\lim_{x\to\infty} \frac{3x^2+\sin(2x)}{x^2+\cos(3x)} = \lim_{x\to\infty} \frac{3x^2+\sin(2x)}{x^2+\cos(3x)} \quad \mathbf{(208)}$$

```
> Limit(numer(f(x)),x=infinity)=limit(numer(f(x)),x=infinity);
  Limit(denom(f(x)),x=infinity)=limit(denom(f(x)),x=infinity);
```

$$\lim_{x\to\infty} \left(3x^2+\sin(2x)\right) = \infty$$

$$\lim_{x\to\infty} \left(x^2+\cos(3x)\right) = \infty \quad \mathbf{(209)}$$

On a une indétermination de la forme $\frac{\infty}{\infty}$. Afin d'appliquer la règle de L'Hospital, formons la limite du quotient des dérivées.

```
> d:=x->Diff(numer(f(x)),x)/Diff(denom(f(x)),x);
```

$$d := x \to \frac{\frac{d}{dx}\, numer(f(x))}{\frac{d}{dx}\, denom(f(x))} \quad \mathbf{(210)}$$

Appliquons le règle de l'Hospital. Développons la limite du quotient des dérivées lorsque $x \to \infty$.

```
> Limit(f(x),x=infinity)=Limit(d(x),x=infinity);
```

$$\lim_{x\to\infty} \frac{3x^2+\sin(2x)}{x^2+\cos(3x)} = \lim_{x\to\infty} \frac{\frac{d}{dx}\left(3x^2+\sin(2x)\right)}{\frac{d}{dx}\left(x^2+\cos(3x)\right)} \quad \mathbf{(211)}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d(x))),x=infinity);
```

$$= \lim_{x\to\infty} -\frac{2\,(3x+\cos(2x))}{-2x+3\sin(3x)} \quad \mathbf{(212)}$$

Analysons de nouveau ce dernier résultat. Évaluons séparément la limite du numérateur et du dénominateur.

```
> Limit(numer((value(d(x)))),x=infinity)=limit(numer((value(d(x)))),
  x=infinity);
  Limit(denom((value(d(x)))),x=infinity)=limit(denom((value(d(x)))),
  x=infinity);
```

$$\lim_{x \to \infty} (-6x - 2\cos(2x)) = -\infty$$

$$\lim_{x \to \infty} (-2x + 3\sin(3x)) = -\infty \qquad \textbf{(213)}$$

On a une indétermination de la forme $\frac{-\infty}{-\infty}$.

Afin d'appliquer de nouveau la règle de L'Hospital, formons encore la limite du quotient des dérivées.

```
> d2:=x->Diff(numer((value(d(x)))),x)/Diff(denom((value(d(x)))),x);
```

$$d2 := x \rightarrow \frac{\frac{d}{dx}\, numer(value(d(x)))}{\frac{d}{dx}\, denom(value(d(x)))} \qquad \textbf{(214)}$$

Appliquons le règle de l'Hospital. Développons la limite du quotient des dérivées lorsque $x \rightarrow \infty$.

```
> Limit(f(x),x=-infinity)=Limit(d2(x),x=-infinity);
```

$$\lim_{x \to -\infty} \frac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} = \lim_{x \to -\infty} \frac{\frac{d}{dx}(-6x - 2\cos(2x))}{\frac{d}{dx}(-2x + 3\sin(3x))} \qquad \textbf{(215)}$$

Calculons les dérivées et simplifions implicitement avec la macro-commande *normal.*

```
> ``=Limit(normal(value(d2(x))),x=infinity);
```

$$= \lim_{x \to \infty} \frac{2(-3 + 2\sin(2x))}{-2 + 9\cos(3x)} \qquad \textbf{(216)}$$

```
> Rule[`c*`]((216));
Creating problem #30 from the right hand side of this equation
```

$$\lim_{x \to \infty} \frac{2(-3 + 2\sin(2x))}{-2 + 9\cos(3x)} = 2\left(\lim_{x \to \infty} \frac{-3 + 2\sin(2x)}{-2 + 9\cos(3x)}\right) \qquad \textbf{(217)}$$

```
> Rule[`/`]((217));
Rule [quotient] does not apply
```

$$\lim_{x \to \infty} \frac{2(-3 + 2\sin(2x))}{-2 + 9\cos(3x)} = 2\left(\lim_{x \to \infty} \frac{-3 + 2\sin(2x)}{-2 + 9\cos(3x)}\right) \qquad \textbf{(218)}$$

Analysons de nouveau ce dernier résultat. Évaluons séparément la limite du numérateur et du dénominateur.

```
> Limit(numer((value(d2(x)))),x=infinity)=limit(numer((value(d2(x)))
  ),x=infinity);
  Limit(denom((value(d2(x)))),x=infinity)=limit(denom((value(d2(x)))
```

```
),x=infinity);
```

$$\lim_{x\to\infty} (-6+4\sin(2x)) = -10..-2$$

$$\lim_{x\to\infty} (-2+9\cos(3x)) = -11..7 \quad \textbf{(219)}$$

La règle quotient ne s'applique donc pas sur $\lim_{x\to\infty} \frac{-3+2\sin(2x)}{-2+9\cos(3x)}$ et on sait maintenant pourquoi.

Et donc, la règle de L'Hospital est alors inefficace pour cette indétermination.

Réécrivons plutôt la limite $\lim_{x\to\infty} \frac{3x^2+\sin(2x)}{x^2+\cos(3x)} = \lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}}$

```
> Rule[`/`](Limit((3+sin(2*x)/x^2)/(1+cos(3*x)/x^2), x = infinity));
Creating problem #31
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = \frac{\lim_{x\to\infty}\left(3+\frac{\sin(2x)}{x^2}\right)}{\lim_{x\to\infty}\left(1+\frac{\cos(3x)}{x^2}\right)} \quad \textbf{(220)}$$

```
> Rule[`+`]((220));
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = \frac{\left(\lim_{x\to\infty} 3\right)+\left(\lim_{x\to\infty}\frac{\sin(2x)}{x^2}\right)}{\lim_{x\to\infty}\left(1+\frac{\cos(3x)}{x^2}\right)} \quad \textbf{(221)}$$

```
> Rule[constant]((221));
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = \frac{3+\left(\lim_{x\to\infty}\frac{\sin(2x)}{x^2}\right)}{\lim_{x\to\infty}\left(1+\frac{\cos(3x)}{x^2}\right)} \quad \textbf{(222)}$$

```
> Rule[`+`]((222));
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = \frac{3+\left(\lim_{x\to\infty}\frac{\sin(2x)}{x^2}\right)}{\left(\lim_{x\to\infty} 1\right)+\left(\lim_{x\to\infty}\frac{\cos(3x)}{x^2}\right)} \quad \textbf{(223)}$$

```
> Rule[constant]((223));
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = \frac{3+\left(\lim_{x\to\infty} \frac{\sin(2x)}{x^2}\right)}{1+\left(\lim_{x\to\infty} \frac{\cos(3x)}{x^2}\right)} \tag{224}$$

```
> Rule[`/`]((224));
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = \frac{3+\left(\lim_{x\to\infty} \frac{\sin(2x)}{x^2}\right)}{1+\frac{\lim_{x\to\infty}\cos(3x)}{\lim_{x\to\infty} x^2}} \tag{225}$$

```
> Rule[`^`]((225));
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = \frac{3+\left(\lim_{x\to\infty} \frac{\sin(2x)}{x^2}\right)}{1+\frac{\lim_{x\to\infty}\cos(3x)}{\left(\lim_{x\to\infty} x\right)^2}} \tag{226}$$

```
> Rule[identity]((226));
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = 3+\left(\lim_{x\to\infty} \frac{\sin(2x)}{x^2}\right) \tag{227}$$

1

```
> Rule[`/`]((227));
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = 3+\frac{\lim_{x\to\infty}\sin(2x)}{\lim_{x\to\infty} x^2} \tag{228}$$

```
> Rule[`^`]((228));
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = 3+\frac{\lim_{x\to\infty}\sin(2x)}{\left(\lim_{x\to\infty} x\right)^2} \tag{229}$$

```
> Rule[identity]((229));
```

$$\lim_{x\to\infty} \frac{3+\frac{\sin(2x)}{x^2}}{1+\frac{\cos(3x)}{x^2}} = 3 \tag{230}$$

L'indétermination est donc levée.

Réponse: $\lim_{x \to \infty} \frac{3x^2 + \sin(2x)}{x^2 + \cos(3x)} = 3$

Exercices

No. 1

Calculer les limites suivantes d'abord directement avec la macro-commande *limit* puis avec un développement pas à pas.

- $\lim_{x \to 0} \frac{x - \tan(x)}{x \sin(x)}$
- $\lim_{x \to 0+} \sqrt{x} \ln(4x)$
- $\lim_{x \to 2+} \left(x^2 - 4\right)^{x-2}$
- $\lim_{x \to \infty} \frac{e^{3x} + 4x - 7}{e^{2x} + 3x - 1}$
- $\lim_{x \to \infty} \frac{x(x+1)(2x+1)}{6x^3}$
- $\lim_{x \to 3+} \left(\frac{3x}{3-x} - \frac{5}{x^2 - 9} \right)$

No. 2

Calculer les limites suivantes d'abord directement avec la macro-commande *limit* puis avec un développement pas à pas.

- $\lim_{x \to \pi^-} \left(\frac{x}{\pi} \right)^{\tan\left(\frac{x}{2}\right)}$
- $\lim_{x \to \infty} x^2 \left(4^{\frac{1}{x}} - 1 \right)$
- $\lim_{x \to \frac{1}{2}^-} \tan^{1-2x}(\pi x)$